

Computing the *Eigenfactor*TM Score and the *Article Influence*TM Score

Carl Bergstrom and Jevin West
University of Washington

Below is the complete source code for the *Eigenfactor* (TM) Algorithm used to compute the *Eigenfactor* (TM) Score and *Article Influence* (TM) Score, using Wolfram Research's *Mathematica* programming language. The three import files are the cross-citation matrix in .mtx sparse matrix format, a list of article counts, and a list of journal names.

```
rawData = Import["Zmatrix2007E.mtx", "MTX"]
articleCount = Import["Article5yr_2007.csv", "CSV"];
journalList = Import["ISIuniqueJrns2007.csv", "CSV"];

zeroDiagonal = rawData - DiagonalMatrix[Diagonal[rawData]];
columnSums = Normal[Apply[Plus, zeroDiagonal]];
danglingNodes = Position[columnSums, 0];
d = ReplacePart[Table[0, {Length[columnSums]}],
  danglingNodes → 1];
cs = ReplacePart[columnSums, danglingNodes → 1];
h = Table[zeroDiagonal[[i, j]] / cs[[j]], {i, 1, Length[cs]},
  {j, 1, Length[zeroDiagonal[[1]]]};
a = articleCount / Apply[Plus, articleCount][[1]];
update[pi_] := .85 h.pi + (.85 (d.pi)[[1]] + .15) a
iter[pi_, k_] := Nest[update, pi, k]
piStar = iter[Table[{1 / Length[articleCount]},
  {Length[articleCount]}], 30]
ef = Module[{prod},
  prod = h.piStar; 100 * prod / Apply[Plus, Flatten[prod]]]
ai = 0.01 * ef / a
resultsTable = Transpose[{Flatten[journalList], Flatten[ef],
  Flatten[ai], Flatten[articleCount]}]

Export["MathematicaEFScores_compressed_2007.csv",
  resultsTable]
```